

TARDIS: Tracking Alterations and Record Differences in SAS, a Macro to Add Color to Your Dataset Comparisons

Inka Leprince, PharmaStat, LLC.

ABSTRACT

Time Lords are an ancient race of time- and space-travelers from the planet Gallifrey. They have taken upon the vital role of safeguarding time-travel technology to protect and preserve order in the universe. The bureaucratic part of this responsibility requires that the High Council of Time Lords keeps a record of events and continually cross-references the current version of events against previous iterations of events. To describe the task of monitoring individuals visiting (or potentially invading) planet Earth—let alone the rest of the universe—as “challenging” would be putting it lightly.

The Tracking Alterations and Record Differences in SAS (TARDIS) macro was developed to ease the High Council's burden by dynamically coloring records to identify antiquated or expunged incidents from the preceding version of events when compared against the current version of events. Additionally, TARDIS applies text formatting to highlight new events that had previously not occurred.

After reading this paper, SAS users will acquire the knowledge to seamlessly integrate the TARDIS macro into existing listing programs, generate color-coded listings, and preserve order in the universe.

INTRODUCTION. CUSTODIANS OF TIME TRAVEL

Time Lords are an ancient race of time- and space-travelers from the planet Gallifrey who act as custodians of time-travel technology to protect and preserve order in the universe. This role involves an extensive bureaucratic responsibility: keeping a record of events and cross-referencing this current version of events against the previous iteration.

To describe the task of monitoring individuals visiting (or potentially invading) planet Earth as “challenging” would be putting it lightly. Yet, the High Council of Time Lords in Gallifrey take upon this burden of routinely reviewing accumulating data from planet Earth. These ongoing reviews mainly rely on reports such as listings.

Color-coded listings are one tool to reduce the workload and increase the efficiency for ongoing data review. Color-coded records highlight the differences between two versions of data and allow reviewers to focus on the changes between the versions. The SAS macro, TARDIS, programmatically applies customized text formats to data records and can be easily integrated within existing listing generation programs.

First, we present how the TARDIS macro fits into the listing generation process. Then, we walk through the SAS macro code as it performs the following functions: 1) records the date stamp of the current data transfer, 2) compares the previously derived listing dataset against the current derived listing dataset, 3) applies customized text formats to color code the new and old records. Finally, we piece all the steps together to produce a historical event listing that has color-coded purple text to identify expunged events from the initial data cut on January 1, 2023, and blue text to identify new records from today's version of events.

WORKFLOW OF CREATING A COLOR-CODED SUBJECT-LEVEL LISTING

Figure 1 shows the workflow to generate a non-color-coded listing output at PharmaStat. The listing program reads in the source data, processes the data, and formats the records into a nice and presentable structure. Listing generation programs produce two outputs: a derived listing SAS dataset, henceforth referred to as AFIL, which is read by the PROC REPORT procedure to also generate a listing output (Figure 2).

Figure 1 Flow Chart of the Listing Generation Process





Figure 2 Sample Listing of Historical Events

Time Lord	Who (Human)	What (Alien Race)	When (Start/End Date)	Where (Planet Earth)	Why (Description)	How (Relevant Alien Tech)
Doctor	Amy Pond	Saturnyns	1580	Venice, Italy	Refugees from The Silence, a family of refugees from Saturnyne fled to 16th century Venice to make a new home for themselves -- while terraforming Venice. The Doctor saves the city but dooms the Saturnyns to extinction.	Perception filters
Doctor	William Shakespeare	Carrionites	1599	Southwark, London, England	Carrionites bewitch Shakespeare to open a portal to their world, their plan is foiled by The Doctor.	Innate power of words Carrionites can use to manipulate psychic energy
Doctor	Charles Dickens	Gelth	1869-12-24/ 1869-12-25	Cardiff, England	Gaseous bodies cross a rift in time and space to reanimate dead bodies, The Doctor extracts the Gelth and sent the Gelth back through the rift.	Rift in time and space
Doctor	Queen Alexandrina Victoria of Kent	Lupine Wavelength Haemovariform	1879	Aberdeenshire, Scotland	Shocked by her close encounter with the supernatural, Queen Victoria forms Torchwood Institute to defend the UK from future alien attacks.	

In order to identify which records have changed between data transfers, several steps need to be taken: 1) retrieve a former version of the derived listing dataset, 2) perform a comparison between the current and former derived listing, and 3) flag or otherwise identify the differences. This process has been consolidated into the TARDIS macro so that this macro call can be seamlessly incorporated inside a listing generation program. Figure 3 displays the process and Figure 6 displays the color-coded listing. The listing created from PROC REPORT can be produced in a number of outputs including: HTML, portable document format (PDF), rich text format (RTF), Microsoft Word, and Microsoft Excel. The TARDIS macro applies text formatting that is compatible with all of these document types.

Figure 3 Flow Chart of a Color-Coded Listing Generation Process with TARDIS



TARDIS COMPONENTS AND FUNCTION

The TARDIS macro performs the following functions to produce color-coded output: 1) records the date of the current data transfer in a permanent dataset, 2) performs a full outer join to compare the records of the previously derived listing dataset against the records in the current derived listing dataset records, 3) programmatically applies a customizable text format to color code the new and old records. In order to perform these functions, the TARDIS macro uses three input datasets and seven macro parameters.

TARDIS INPUTS THREE DATASETS

1. **TLGDATA.TransferDates**: a dataset that stores the date stamps of data transfers. It contains 4 variables:

Table 1 TransferDates Variables

Variable Name	Variable Label	Type	Format	Length
DataDate	Date of Current Data Transfer	Integer	YYMMDD10.	8
PrevDataDate	Date of Previous Data Transfer	Integer	YYMMDD10.	8
TimeStamp	Date Stamp of Current Data Transfer	Char	\$9	9
PrevStamp	Date Stamp of Previous Data Transfer	Char	\$9	9


Note: This dataset can also be configured to store other variables, for example, the run date of the listing, or the date when an intermediary dataset (e.g., Study Data Tabulation Model [SDTM] or Analysis Data Model [ADaM]) was updated.

- Date of Current Data Transfer (integer, date), *This is the date last modified of the dataset specified within the TARDIS macro.* In the sample code (Appendix A: TARDIS.SAS), the SAS macro grabs the date last modified from the raw, source dataset, EARTH.sas7bdat, that is provided by the High Council of Time Lords (Appendix B3: EARTH).
- Date of Previous Data Transfer (integer, date), *This is the date when the previous version of data files was transferred, derived using the lag function in a DATA step.*

- Date Stamp of Current Data Transfer (character, \$9) *This is the character value of the Date of Current Data Transfer, preceded by an underscore. These values follow the format “_YYYYMMDD”*
- Date Stamp of Previous Data Transfer (character, \$9) *This the character value of the Date of Previous Data Transfer, preceded by an underscore. These values follow the format “_YYYYMMDD”*

Using the sample SAS code in Appendix B1: TransferDates, the TLGDATA.TransferDates dataset can be created with the initial value of the first observation with a TimeStamp = 'NONE', and a second record containing the time stamp of January 1, 2023, the date of the first data cut we received of Earth data from the High Council of Time Lords.


Figure 4 TransferDates (Date of Transfer Dataset)



	TimeStamp	PrevStamp	DataDate	PrevDataDate
1	NONE		.	.
2	_20230101	NONE	2023-01-01	.
3	_20231031	_20230101	2023-10-31	2023-01-01

2. AFILE: the derived dataset that supports the listing, which is produced from the current data transfer. This is the dataset that is usually read by the PROC REPORT procedure, but instead will be input and processed by the TARDIS macro. In the sample code, AFILE is derived from the Earth's historical event dataset (Appendix B3: EARTH) in the listing code, Appendix C: .
3. TIMEDATA.&BASEPROG.&ORDER.&PREVSTAMP: a derived listing dataset that supports the listing, which was produced from the previous data transfer. The sample SAS code in Appendix B2: L_HISTORY_EARTH_20230101 provides L_HISTORY_EARTH_20220101, a derived listing dataset for historical events with an artificially assigned date stamp of January 1, 2022 can be created with initial values (see image below for example).

Figure 5 Derived History Listing Data Set with Date Stamp 2023-01-01



	TIMELORD	HUMAN_C	ALIEN_RACE	DATE_C	LOCATION_C	EVENT	ALIEN_TECH
1	Doctor	Amy Pond	Saturnyns	1580	Venice, Ital	Refugees fro	Perception f
2	Doctor	Charles Dickens	Gelth	1869-12-24/~n1869	Cardiff, Eng	Gaseous bodi	Rift in time
3	Doctor	Queen Alexandrin	Lupine Wavele	1879/~n1901-01-22	Aberdeenshir	Queen Victor	
4	Doctor	Adolf Hitler		1938/~n5100	Berlin, Germ	The Doctor p	Teselecta
5	Doctor	Captain Jack Har	Chula	1941-01	London, Engl	During the B	Chula medica
6	Doctor	Rose Tyler	Nestene Consc	2005-03-04/~n2005	London, Engl	Rose Tyler s	London Eye a
7	Doctor	Sally Sparrow	Weeping Angel	2007/~n1920	Hull, Englan	A small grou	Innate abili
8	Missy	Danny Pink	Cybermen	2014-11-01/~n2014	London, Engl	Missy releas	Nethersphere
9	Doctor	Nasreen Chaudhry	Silurians	2020	Cwmtaff, Wal	A drilling c	
10	Doctor	Rose Tyler	Sisters of Pl	5000000023	New New New	The Sisters	

TARDIS HAS SEVEN MACRO PARAMETERS

The first three parameters (&INDATA, &BASEPROG, &ORDER) are used to identify, save, and retrieve the appropriately named derived listing datasets. The following two parameters (&NEWCOLOR and &PREVCOLOR) allow the users to easily customize the text formats applied to identify new and old records, and &ARRAYVAR specifies which character variables—corresponding to the columns in the output listings--will have the text formatting applied to them. &SORTVAR is used for data comparison when merging the derived listing dataset with its predecessor.

Table 2 TARDIS Macro Parameters

	Parameter	Valid Value	Sample Value	Description
1.	INDATA	Dataset name	INDATA = AFILE	Input dataset name, the derived dataset that supports the listing to be generated
2.	BASEPROG	text	BASEPROG = L_HISTORY	The name of the base listing program
3.	ORDER	text	ORDER = _EARTH	The order of the listing program, may have values to indicate the subsets or variations of the base programs
4.	NEWCOLOR	text	NEWCOLOR = `S={color=blue}	The text format to be applied to new/changed data from the current data transfer. The default is set to blue font color
5.	PREVCOLOR	text	PREVCOLOR = `S={textdecoration=line_through color=purple}	The text format to be applied original/deleted data from the previous data transfer. The default is set to a purple font color with strikethrough
6.	SORTVAR	Variable names	SORTVAR = styear astdt enyear aendt timelord human_c alien_race date_c location_c event alien_tech	The list of variable names, delimited by a space, for the sort and merging order of AFILE with its timestamped predecessor. This list typically contains all the variables that are used in the COLUMN statement in PROC REPORT
7.	ARRAYVAR	Character variable names	ARRAYVAR = timelord human_c alien_race date_c location_c event alien_tech	The list of variable names, delimited by a space, for the character variables whose values will be formatted with either the new or previous text formatting

TARDIS OUTPUTS TWO DATASETS

1. TLGDATA.TransferDates: a dataset that stores the date stamps of data transfers, updated to include the current date stamp.
2. TIMEDATA.&BASEPROG.&ORDER.&TIMESTAMP: AFILE saved in the TimeData folder with a date stamp of the current data transfer. This dataset will be used for future comparisons. Its values do not have any text formatting.
3. WORK.AFILE: After permanently saving AFILE, the TARDIS macro updates and outputs the temporary dataset, AFILE, which now has colored text formatting applied to new and old records. This is the dataset that will be read into the PROC REPORT procedure. The TARDIS macro applies the following changes

Table 3 AFILE New Variables Created by TARDIS

Variable Name	Variable Label	Type	Format	Length
NewDataFL	New Data Record Flag	Char	\$1	1
ColorData	Apply Color/Text Formatting to this Record?	Char	\$3	3
Data_Color	Color/Text Formatting	Char	\$50	50

- New Data Record Flag (character, \$1), *This is set to 'Y' if a record in AFILE has new data, null otherwise.* This flag can be used in PROC REPORT or additional data steps for extra processing of new data.
- Apply Color/Text Formatting to this Record? (character, \$3), *This is set to 'YES' to identify that the character variables identified in &ARRAYVAR should be assigned color/text formatting.*

- Color/Text Formatting (character, \$50) *This variable is assigned values from &NEWCOLOR or &PREVCOLOR, as applicable.* For records where ColorData = 'YES', all of the variables identified in &ARRAYVAR will have the text formatting specified by Data_Color appended to the beginning of their values.

TARDIS MACRO FUNCTION

This macro is designed to be called by a listing program right before the PROC REPORT procedure. The derived listing dataset, AFILE, that is normally read in by the PROC REPORT will instead be processed by the TARDIS macro, which will compare the records from AFILE to the records generated from a previous iteration of the listing program. TARDIS will apply text formats to character variables in AFILE to color-code changes in the data.

This macro consists two parts: time stamping and data comparison.

CREATING TIME STAMPED VARIABLES

In order for the TARDIS macro to run with the sample data provided in Appendix C: , the macro is currently designed to go into the Source Data directory to retrieve a date (e.g., the date of the EARTH.sas7bdat file was last updated). This date is saved in the macro variable, RAWDATE:

```
%let rc      = %sysfunc(filename(onefile, &srcdata\earth.sas7bdat)) ;
%let fid     = %sysfunc(fopen(&onefile)) ;
%let rawdate = %substr(%qsysfunc(finfo(&fid,Last Modified)),1,9) ;
%let fdc     = %sysfunc(fclose(&fid)) ;
```

The first DATA step creates a temporary dataset, NewSaved, with one observation and two variables DATADATE (a numeric variable containing the SAS date of RAWDATE) and TIMESTAMP (a character variable storing digits of RAWDATE preceded by an underscore). Additionally, the macro variable TIMESTAMP is defined and will be used to save date stamped versions of AFILE, the derived listing dataset.

```
data NewSaved ;
  length TimeStamp $9. ;
  DataDate = input("&rawdate", date9.) ;
  TimeStamp = '_' || compress(put(DataDate, yymmdd10.), '-') ;
  call symput('timestamp', timestamp);
  format datadate yymmdd10.;
run ;
```

The second DATA step is designed to record the unique values of DATADATE in TransferDates. Datasets WORK.NewSaved and TLGDATA.TransferDates are merged by DATADATE in a full outer join. If DATADATE is a new date that is not contained in the TLGDATA.TransferDates, it creates a new record in TLGDATA.TransferDates. If the DATADATE already exists in TLGDATA.TransferDates, then no changes are made. In this data step, we also define the two new variables PREVDATADATE and PREVSTAMP, which holds the YYMMDD10. and character values, respectively, of the previously stored value of DATADATE. These two macro variables will aid in the comparison and saving of date stamped derived datasets. The value of PREVSTAMP is also stored in the macro variable, PREVSTAMP, in order to be able to call the derived listing made from the previous date:

```
data tlgdata.TransferDates ;
  merge tlgdata.TransferDates (in=inSavedDates)
        newSaved ;
  by DataDate ;

  lagPrevDataDate = lag(DataDate) ;
  lagPrevStamp = lag(TimeStamp) ;

  if not inSavedDates then do ;
    PrevDataDate = lagprevDataDate ;
    PrevStamp = lagprevStamp ;
  end ;

  call symput('prevstamp',prevstamp);
  drop lagprevStamp lagprevDataDate ;
run ;
```

COMPARING TWO VERSIONS OF DATA

The second part of the macro indicates new or changed data within the derived dataset that will be used to create the listing. A data saves AFILF with the timestamp of TIMESTAMP and stores the dataset in the TimeData folder for future comparisons:

```
data timedata.&baseprog.&order.&timestamp ;
    set &indata ;
    keep &sortvar ;
run ;
```

If there is no previous record (&PREVSTAMP= NONE) then the macro passes the dataset through without adding any color text formats to the data because the previous version of the dataset does not exist and would cause the MERGE step to fail.

Otherwise, after saving AFILF, the previous (TIMEDATA.&BASEPROG.&ORDER.&PREVSTAMP) version of the derived listing dataset and the current version (AFILF) of the derived listing dataset are merged by the variables listed in &SORTVAR in an full outer join. Three new variables, NewDataFL, ColorData, and Data_Color, are defined (Table 3) in this data step as well. Any records that do not fall within the inner join of the two datasets will be flagged for text formatting (ColorData = 'YES'). If the records are not in the previously derived dataset, then it must be "new" data, and is flagged NewDataFL = 'Y' and marked for text formatting ColorData = 'YES', and the Data_Color is set to &NEWCOLOR. If the records are not in the new data then they must have been changed or removed. In this case, the previous data records are also marked for text formatting ColorData = 'YES' and the Data_Color is set to the &PREVCOLOR. Then, a DO Loop is called and goes through the array of character variables defined in &ARRAYVAR and it appends the text formatting at the beginning of the character value of each of the array variables, when ColorData = 'YES':

```
data &indata ;
%if &prevstamp ^= NONE %then %do ;
    length &arrayvar $250 data_color $50 ;
    merge timedata.&baseprog.&order.&prevstamp (in=inprev)
          &indata (in=innew) ;
    by &sortvar ;

    /* if not in previous data, then must be new data */
    if not inprev then do ;
        newdatafl = 'Y' ;
        colordata = 'YES' ;
        data_color = "&newcolor" ;
    end ;

    /* if not in new data, then be removed/changed records in previous data */
    if not innew then do ;
        colordata = 'YES' ;
        data_color = "&prevcolor" ;
    end ;

    array cols &arrayvar ;
    if colordata='YES' then do ;
        do over cols ;
            cols = cats(data_color,cols) ;
        end ;
    end ;
%end ;
%else %do ;
    set &indata ;
%end ;
run ;
```

The text formatting is appended to the beginning of the values of the variables specified by the macro parameter &ARRAYVAR, and these values are output in an updated AFILF. Once this temporary dataset is read by PROC REPORT, the colorized historical event listing shown in Figure 6 is produced.

COLOR-CODED LISTING OUTPUT

Figure 6 demonstrates how the colored output can be used to identify changed, new, and deleted records. Since the data was last updated in January 2023, Doctor Who has had a new interaction with William Shakespeare and helped save planet Earth from an invasion of Carrionites, as identified by the blue text. The third historical event concerning Queen Victoria can be identified as a changed record because there is a new record in blue text immediately followed by the previous record in purple strikethrough text that describes basically the same event. As of the current version of events, Queen Victoria did not succumb to the Lupine Wavelength Haemovariform and instead established Torchwood Institute. Previously, Queen Victoria was infected by this alien “werewolf” bacteria/virus that affected her life and the path of the United Kingdom until the day she died, Jan 22, 1901. It also appears that the Time Lord Missy opted not to build army of Cybermen, so that event has since been deleted from the Earth’s history, as identified by the purple strikethrough text.

Figure 6 Sample Color-Coded Listing of Historical Events

Time Lord	Who (Human)	What (Alien Race)	When (Start/End Date)	Where (Planet Earth)	Why (Description)	How (Relevant Alien Technology)
Doctor	Amy Pond	Saturnyns	1580	Venice, Italy	Refugees from The Silence, a family of refugees from Saturnyne fled to 16th century Venice to make a new home for themselves -- while terraforming Venice. The Doctor saves the city but dooms the Saturnyns to extinction.	Perception filters
Doctor	William Shakespeare	Carrionites	1599	Southwark, London, England	Carrionites bewitch Shakespeare to open a portal to their world, their plan is foiled by The Doctor.	Innate power of words Carrionites can use to manipulate energy New
Doctor	Charles Dickens	Gelth	1869-12-24/ 1869-12-25	Cardiff, England	Gaseous bodies cross a rift in time and space to reanimate dead bodies, The Doctor extracts the Gelth and sent the Gelth back through the rift.	Rift in time and space
Doctor	Queen Alexandrina Victoria of Kent	Lupine Wavelength Haemovariform	1879	Aberdeenshire, Scotland	Shocked by her close encounter with the supernatural, Queen Victoria forms Torchwood Institute to defend the UK from future alien attacks.	Changed
Doctor	Queen Alexandrina Victoria of Kent	Lupine Wavelength Haemovariform	1879/ 1901-01-22	Aberdeenshire, Scotland	Queen Victoria becomes a werewolf (is infected by an alien spore/virus whose symptomology mimics a humanoid canine) and establishes an Empire of the Wolf.	
Doctor	Rose Tyler	Nestene Consciousness	2005-03-04/ 2005-03-05	London, England	Rose Tyler stops the the Nesene consciousness from invading planet Earth by killing it with a vial of anti-plastic.	London Eye as a consciousness transmitter
Doctor	Sally Sparrow	Weeping Angels	2007/ 1920	Hull, England	A small group of weeping angels prey on humans' time-potential energy. With The Doctor's help, these weeping angels are tricked into looking at each other and are quantum locked as statues forever.	Innate ability of Weeping Angels to displace beings in time
Missy	Danny Pink	Cyberman	2014-11-01/ 2014-11-08	London, England	Missy releases cybermen from the globally distributed 3W Institute mausoleums. Danny Pink takes control of all Cyberman and commands all Cybermen to stand down and dismantle the Nethersphere.	Nethersphere Matrix Deleted
Doctor	Nasreen Chaudhry	Silurians	2020	Cwmtaff, Wales	A drilling company discovers an underground civilization of ancient reptilian humanoids. The Doctor mediates talks and prevents all out war between the Silurians and the humans.	
Doctor	Rose Tyler	Sisters of Plenitude	5000000023	New New New New New New New New New New New New New York, United States of America	The Sisters of Plenitude were farming human clones for medical advancement. The Doctor frees the clones and establishes their citizenship.	

CUSTOMIZATIONS

The text formatting can be customized for old/changed data and new/current data. These text formats include strikethrough, highlights, bold, italics, colors, and more. Please refer to SAS documentation, SAS® 9.4 Output Delivery System: Advanced Topic and Eric Gerbhart's SAS Global Forum Paper 222-2009, on SAS text formats.

If the previous old/changed records are not the review focus, the previous records can easily be suppressed by implementing the following SAS code:

```
/* if not in previous data, then must be new data */
if not inprev then do ;
    newdatafl = 'Y' ;
    colordata = 'YES' ;
    data_color = "&newcolor" ;
end ;

/* if not in new data, then must be removed/changed records in previous data */
if not innew then delete ;
```

If there is a set of listings that utilize the same source data and are always produced and delivered at the time, this set of listings can all use the same TransferDates dataset instead of creating a unique TLGDATA.TransferDates dataset to be used by each listing program. It is important to note that all the listing programs associated with a TLGDATA.TransferDates dataset should be run at least once per data transfer in order to log and store the timestamped derived listing datasets of all the listings from each data transfer.

CAUTIONARY TALES

One weakness of this TARDIS macro is that it is not flexible to changes in the derived dataset that supports the listing.

If, for example, the High Council wants to add a new column such as a flag for how severe the repercussions of an event was, the macro will crash. **Workaround:** Update the PROC REPORT to include the new variable, but do not include the new variable in the TARDIS keyword parameters &SORTVAR and &ARRAYVAR values until the next data transfer, when the new variable becomes available in the previous and current datasets.

If you modify one or more of the values in the listings (e.g., displaying the human names by last name and then first name), all affected observations will be flagged as new records. Workaround: implement these updates *prior* to the data transfer.

CONCLUSION

With only minor updates to the environment and listing macros, the TARDIS macro can be easily implemented to generate colorized listings that highlight the differences between two sets of source data. Modifications can be made to the macro to customize the text formats to add strikethrough, bold, italicize, highlighting, etc. A SAS user can also preferentially display only new records, only old records, or both in the color-coded subject listings.

REFERENCES

- Gerbhart, E. (2009). *Paper 222-2009 Inline Formatting with ODS Markup*. SAS Global Forum. <https://support.sas.com/resources/papers/proceedings09/222-2009.pdf>
- Leprince, I., Li, E., Chesbrough, C. (2022) *Paper 134: TrackCHG: A SAS Macro to Colorize and Track Changes Between Data Transfers in Subject-Level Safety Listings*. WUSS. <https://www.wuss.org/proceedings/2022/WUSS-2022-Paper-134.pdf>
- McCandless, D. (2010, August 10). *Doctor Who: Every Single Journey through Time Detailed by Information Is Beautiful*. The Guardian. www.theguardian.com/news/datablog/2010/aug/20/doctor-who-time-travel-information-is-beautiful.

SAS Institute Inc. (2016). *SAS® 9.4 Output Delivery System: Advanced Topics, Third Edition*. Cary, NC: SAS Institute Inc. (pp. 216-220, 253–259).

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Inka Leprince
PharmaStat, LLC
(510) 709-9107
ileprince@pharmastat.com

APPENDIX A: TARDIS MACRO CODE

```
** ----- ** ;
** Program:      TARDIS.sas ;
** Author:       IMLeprince ;
** Created:      2023-10-12 ;
** Inputs:       tlfdata.TransferDates  Contains timestamps of data transfers ;
**               timedata.&baseprog&order.&prevstamp ;
**               AFILE created from the previous ;
**               (&prevstamp) data transfer ;
**               work.afile              Analysis file for listing ;
**               ;
** Outputs:      tlfdata.TransferDates  Contains timestamps of data transfers ;
**               timedata.l_baseprog&order.&timestamp ;
**               AFILE saved with a timestamp ;
**               (&timestamp) of current data transfer ;
**               work.afile              Analysis file for listing with color- ;
**               coded records ;
**               ;
** Usage:        Track Changes between current vs previous data transfer ;
**               for data listings ;
** Revisions:    ;
** ----- ** ;

%macro TARDIS (
    indata      = ,
    baseprog    = ,
    order       = ,
    newcolor    = `S={foreground=blue},
    prevcolor   = `S={textdecoration=line_through foreground=purple},
    sortvar     = ,
    arrayvar    =
) ;

%*-----;
%* Save most current data transfer date into TransferDates ;
%* Create macro variables: ;
%* (1) Timestamp: YYYYMMDD of current data transfer ;
%* (2) PrevStamp: YYYYMMDD of previous data transfer ;
%*-----;
```

```

/* Preference to get the date of data transfer from the raw/source dataset provided from data vendor */
/* For utilization with example Historiy listing, getting date of data transfer from EARTH.sas7bdat */
%let rc      = %sysfunc(filename(onefile, &srcdata\earth.sas7bdat)) ;
%let fid     = %sysfunc(fopen(&onefile)) ;
%let rawdate = %substr(%qsysfunc(finfo(&fid,Last Modified)),1,9) ;
%let fidc    = %sysfunc(fclose(&fid)) ;

data NewSaved ;
  length TimeStamp $9. ;
  DataDate = input("&rawdate", date9.) ;

  TimeStamp = '_' || compress(put(DataDate, yymmdd10.), '-') ;
  call symput('timestamp', timestamp);
  format datadate yymmdd10. ;
run ;
%put &timestamp ;

/* Record date of current data transfer in tlfdata.TransferDates */
%if %sysfunc(exist(tlfdata.TransferDates)) %then %do;
  data tlfdata.TransferDates ;
    merge tlfdata.TransferDates (in=inSavedDates)
          newSaved ;
    by DataDate ;

    lagPrevDataDate = lag(DataDate) ;
    lagPrevStamp = lag(TimeStamp) ;

    if not inSavedDates then do ;
      PrevDataDate = lagprevDataDate ;
      PrevStamp = lagprevStamp ;
    end ;

    call symput('prevstamp',prevstamp);
    drop lagprevStamp lagprevDataDate ;
  run ;
%end;
%else %if %sysfunc(exist(tlfdata.TransferDates))=0 %then %do ;
  data tlfdata.TransferDates ;
    format TimeStamp PrevStamp $9. DataDate PrevDataDate YYMMDD10. ;
    label TimeStamp = "Date Stamp of Current Data Transfer"

```

```

PrevStamp = "Date Stamp of PRevious Data Transfer"
DataDate  = "Date of Current Data Transfer"
PrevDataDate = "Date of Previous Data Transfer";

PrevDataDate = . ;
PrevStamp = 'NONE' ;

call symput('prevstamp',prevstamp);
run ;
%end ;

%*-----;
%* Indicate New or Changed Data within Listing
%*-----;
/* Sort input data by &SORTVARS */
proc sort data = &indata ;
    by &sortvar ;
run ;

/* Save a timestamped copy of the current dataset for future comparison */
data timedata.&baseprog.&order.&timestamp ;
    set &indata ;
    keep &sortvar ;
run ;

/* Merge previous and current derived datasets */
/* Previous data transfer: timedata.&baseprog.&order.&prevstamp */
/* Current data transfer: &indata */
data &indata ;

%if &prevstamp ^= NONE %then %do ;
    length &arrayvar $250 data_color $50 ;
    merge timedata.&baseprog.&order.&prevstamp (in=inprev)
           &indata (in=innew) ;
    by &sortvar ;

    /* if not in previous data, then must be new data */
    if not inprev then do ;
        newdatafl = 'Y' ;
        colordata = 'YES' ;
    end ;
%end ;

```

```

        data_color = "&newcolor" ;
    end ;

    /* if not in new data, then be removed/changed records in previous data */
    if not innew then do ;
        colordata = 'YES' ;
        data_color = "&prevcolor" ;
    end ;

    array cols &arrayvar ;
    if colordata='YES' then do ;
        do over cols ;
            cols  = cats(data_color,cols) ;
        end ;
    end ;
%end ;
%else %do ;
    set &indata ;
%end ;
run ;

%mend ;

```

APPENDIX B: SAMPLE DATA

```

%*-----;
%* Setting up environment: define the filepaths
%*-----;
/* SRCDATA   = where the source dataset (EARTH.sas7bdat) resides */
/* TLFDATA   = where the derived datasets that support outputs resides */
/* TIMEDATA  = where the timestamped derived datasets that support */
/*           outputs resides */

%let SRCDATA   = C:\Users\ileprince\Desktop\BASUG 2025\TLFs\Source Data ;
%let TLFDATA   = C:\Users\ileprince\Desktop\BASUG 2025\TLFs\Derived Data ;
%let TIMEDATA  = C:\Users\ileprince\Desktop\BASUG 2025\TLFs\Derived Data\TimeStamped ;

libname srcdata "&srcdata" ;
libname tlfdata "&tlfdata" ;
libname timedata "&timedata" ;

```

APPENDIX B1: TRANSFERDATES.SAS7BDAT

```
%*-----;
%* TransferDates: Date of Transfer Saved
%* Stores the date stamps of data transfers,
%* Contains record for initial data received on 2025-01-01
%*-----;
data tlfdata.TransferDates ;
    infile datalines missover ;
    informat Timestamp PrevStamp $9. DataDate PrevDataDate YYMMDD10. ;
    input Timestamp PrevStamp $ DataDate PrevDataDate ;
    label Timestamp = "Date Stamp of Current Data Transfer"
           PrevStamp = "Date Stamp of Previous Data Transfer"
           DataDate  = "Date of Current Data Transfer"
           PrevDataDate = "Date of Previous Data Transfer";
    format DataDate PrevDataDate YYMMDD10. ;
    datalines ;
NONE
_20250101 NONE      2025-01-01
;
run ;
```

APPENDIX B2: L_HISTORY_EARTH_20250101.SAS7BDAT

```
%*-----;
%* l_history_earth_20250101: derived dataset of history listing
%*                          from the 2025-01-01 data transfer
%*-----;
data timedata.l_history_earth_20250101 ;
    infile datalines dlm = '*' dsd ;
    input TIMELORD :$6. C_HUMAN :$200. ALIEN_RACE :$35.
           C_DATE :$200. STYEAR :12. ASTDT :YYMMDD10. ENYEAR :12. AENDT :YYMMDD10.
           C_LOCATION :$200. EVENT :$250. ALIEN_TECH :$75. ;
    format ASTDT YYMMDD10. AENDT YYMMDD10. ;
    label TIMELORD = 'Time Lord'
           C_HUMAN = 'Who (Human)'
           ALIEN_RACE = 'Alien Race'
           C_DATE = 'When (Start/End Date)'
           STYEAR = 'Start Year'
           ASTDT = 'Analysis Start Date'
           ENYEAR = 'End Year'
           AENDT = 'Analysis End Date'
```

```

C_LOCATION = 'Where (Planet Earth)'
EVENT = 'Event Description'
ALIEN_TECH = 'Alien Technology' ;

datalines ;
Doctor*Amy Pond*Saturnyns*1580*1580*.*1580*.*Venice, Italy*Refugees from The Silence, a family fled from Saturnyne to 16th century Venice to make a new home for themselves -- while terraforming Venice. The Doctor saves the city but dooms the Saturnyns to extinction.*Perception filters
Doctor*Charles Dickens*Gelth*1869-12-24/\`n1869-12-25*1869*1869-12-24*1869*1869-12-25*Cardiff, England*Gaseous bodies cross a rift in time and space to reanimate dead bodies, The Doctor extracts the Gelth and sent the Gelth back through the rift.*Rift in time and space
Doctor*Queen Alexandrina Victoria of Kent*Lupine Wavelength Haemovariform*1879/\`n1901-01-22*1879*1879-01-01*1901*1901-01-22*Aberdeenshire, Scotland*Queen Victoria becomes a werewolf (is infected by an alien spore/virus whose symptomology mimics a humanoid canine) and establishes an Empire of the Wolf.*
Doctor*Adolf Hitler**1938/\`n5100*1938*1938-01-01*5100*5100-01-01*Berlin, Germany*The Doctor prevents a rip in the time and space continuum by saving Hitler from being assassinated by a vigilante human crew from the future.*Teselecta
Doctor*Captain Jack Harkness*Chula*1941-01*1941*1941-01-01*1941*1941-01-01*London, England*During the Blitz, an alien ship crashes in London. The alien nanogene technology attempts to heal surrounding human bodies but turns the wounded into zombies seeking their mother, "Are you my mummy?" The Doctor aids in reprogramming the nanobots to cure the London population.*Chula medical nanobots
Doctor*Rose Tyler*Nestene Consciousness*2005-03-04/\`n2005-03-05*2005*2005-03-04*2005*2005-03-05*London, England*Rose Tyler stops the the Nesene consciousness from invading planet Earth by killing it with a vial of anti-plastic.*London Eye as a consciousness transmitter
Doctor*Sally Sparrow*Weeping Angels*2007/\`n1920*2007*2007-01-01*1920*1920-01-01*Hull, England*A small group of weeping angels prey on humans' time-potential energy. With The Doctor's help, these weeping angels are tricked into looking at each other and are quantum locked as statues forever.*Innate ability of Weeping Angels to displace beings in time
Missy *Danny Pink*Cybermen*2014-11-01/\`n2014-11-08*2014*2014-11-01*2014*2014-11-08*London, England*Missy releases cybermen from the globally distributed 3W Institute mausoleums. Danny Pink takes control of all Cyberman and commands all Cybermen to stand down and dismantle the Nethersphere.*Nethersphere Matrix
Doctor*Nasreen Chaudhry*Silurians*2020*2020*2020-01-01*2020*2020-01-01*Cwmataff, Wales*A drilling company discovers an underground civilization of ancient reptilian humanoids. The Doctor mediates talks and prevents all out war between the Silurians and the humans.*
Doctor*Rose Tyler*Sisters of Plenitude*5000000023*5000000023*.*5000000023*.*New New New New New New New New New New New New New New New New York, United States of America*The Sisters of Plenitude were farming human clones for medical advancement. The Doctor frees the clones and establishes their citizenship.*
;

run ;

```

APPENDIX B3: EARTH.SAS7BDAT

```
%*-----;
%* Earth: Dataset of historical events on Earth
%* Current data transfer
%*-----;

data SRCDATA.earth ;
  infile datalines dlm = '*' dsd truncover ;
  input EPISODE :$25. TIMELORD :$6. STDATEC :$10. ENDATEC :$10.
        THREAT :$10. PLANET :$9. COUNTRY :$24. CITY :$64.
        HUMAN_LAST :$21. HUMAN_FIRST :$21. HUMAN_MIDDLE :$21. HUMAN_TITLE :$21.
        ALIEN_RACE :$31. ALIEN_TECH :$75. EVENT :$250. ;
  label EPISODE      = "Doctor Who Episode"
        TIMELORD      = "Time Lord Name"
        STDATEC       = "Start Date"
        ENDATEC       = "End Date"
        THREAT        = "Disaster Level"
        PLANET        = "Planet"
        COUNTRY       = "Country"
        CITY          = "City"
        HUMAN_LAST    = "Human Last Name"
        HUMAN_FIRST   = "Human First Name"
        HUMAN_MIDDLE  = "Human Middle Name"
        HUMAN_TITLE   = "Human Title"
        ALIEN_RACE    = "Alien Race"
        ALIEN_TECH    = "Alien Technology"
        EVENT         = "Event Description"
        ;
  datalines ;
Rose*Doctor*2005-03-04*2005-03-05*Wolf*Earth*England*London*Tyler*Rose***Nestene Consciousness*London Eye
as a consciousness transmitter*Rose Tyler stops the the Nesene consciousness from invading planet Earth by
killing it with a vial of anti-plastic.
The Unquiet Dead*Doctor*1869-12-24*1869-12-25*Dragon*Earth*England*Cardiff*Dickens*Charles***Gelth*Rift in
time and space*Gaseous bodies cross a rift in time and space to reanimate dead bodies, The Doctor extracts
the Gelth and sent the Gelth back through the rift.
The Empty Child*Doctor*1941-01*1941-01*Tiger*Earth*England*London*Harkness*Jack**Captain*Chula*Chula
medical nanobots*During the Blitz, an alien ship crashes in London. The alien nanogene technology attempts
to heal surrounding human bodies but turns the wounded into zombies seeking their mother, "Are you my
mummy?" The Doctor aids in reprogramming the nanobots to cure the London population.
New Earth*Doctor*5000000023*5000000023*Tiger*New Earth*United States of America*New New New New New New
New New New New New New New York*Tyler*Rose***Sisters of Plenitude**The Sisters of Plenitude were
```

farming human clones for medical advancement. The Doctor frees the clones and establishes their citizenship.

Tooth and Claw*Doctor*1879*1879*Wolf*Earth*Scotland*Aberdeenshire*of Kent*Alexandrina*Victoria*Queen*Lupine Wavelength Haemovariform**Shocked by her close encounter with the supernatural, Queen Victoria forms Torchwood Institute to defend the UK from future alien attacks.

The Shakespeare Code*Doctor*1599*1599*God*Earth*England*Southwark, London*Shakespeare*William***Carrionites*Innate power of words Carrionites can use to manipulate psychic energy*Carrionites bewitch Shakespeare to open a portal to their world, their plan is foiled by The Doctor. Blink*Doctor*2007*1920*Wolf*Earth*England*Hull*Sparrow*Sally***Weeping Angels*Innate ability of Weeping Angels to displace beings in time*A small group of weeping angels prey on humans' time-potential energy. With The Doctor's help, these weeping angels are tricked into looking at each other and are quantum locked as statues forever.

The Vampires of Venice*Doctor*1580*1580*Demon*Earth*Italy*Venice*Pond*Amy***Saturnyns*Perception filters*Refugees from The Silence, a family fled from Saturnyne to 16th century Venice to make a new home for themselves -- while terraforming Venice. The Doctor saves the city but dooms the Saturnyns to extinction.

The Hungry Earth*Doctor*2020*2020*God*Earth*Wales*Cwmataff*Chaudhry*Nasreen***Silurians**A drilling company discovers an underground civilization of ancient reptilian humanoids. The Doctor mediates talks and prevents all out war between the Silurians and the humans.

Let's Kill Hitler*Doctor*1938*5100*Wolf*Earth*Germany*Berlin*Hitler*Adolf****Teselecta*The Doctor prevents a rip in the time and space continuum by saving Hitler from being assassinated by a vigilante human crew from the future.

```
;  
run ;
```

```
/* No need to run, but this is "Source" Earth data from for 20250101 transfer */  
data earth_20250101 ;  
  infile datalines dlm = '*' dsd truncover ;  
  input EPISODE :$25. TIMELORD :$6. STDATEC :$10. ENDATEC :$10.  
    THREAT :$10. PLANET :$9. COUNTRY :$24. CITY :$64.  
    HUMAN_LAST :$21. HUMAN_FIRST :$21. HUMAN_MIDDLE :$21. HUMAN_TITLE :$21.  
    ALIEN_RACE :$31. ALIEN_TECH :$75. EVENT :$250. ;  
  label EPISODE      = "Doctor Who Episode"  
    TIMELORD         = "Time Lord Name"  
    STDATEC          = "Start Date"  
    ENDATEC          = "End Date"  
    THREAT           = "Disaster Level"  
    PLANET           = "Planet"  
    COUNTRY          = "Country"  
    CITY             = "City"
```

```

HUMAN_LAST      = "Human Last Name"
HUMAN_FIRST     = "Human First Name"
HUMAN_MIDDLE    = "Human Middle Name"
HUMAN_TITLE     = "Human Title"
ALIEN_RACE      = "Alien Race"
ALIEN_TECH      = "Alien Technology"
EVENT           = "Event Description"

```

```
;
```

```
datalines ;
```

```

Rose*Doctor*2005-03-04*2005-03-05*Wolf*Earth*England*London*Tyler*Rose***Nestene Consciousness*London Eye
as a consciousness transmitter*Rose Tyler stops the the Nesene consciousness from invading planet Earth by
killing it with a vial of anti-plastic.
The Unquiet Dead*Doctor*1869-12-24*1869-12-25*Dragon*Earth*England*Cardiff*Dickens*Charles***Gelth*Rift in
time and space*Gaseous bodies cross a rift in time and space to reanimate dead bodies, The Doctor extracts
the Gelth and sent the Gelth back through the rift.
The Empty Child*Doctor*1941-01*1941-01*Tiger*Earth*England*London*Harkness*Jack**Captain*Chula*Chula
medical nanobots*During the Blitz, an alien ship crashes in London. The alien nanogene technology attempts
to heal surrounding human bodies but turns the wounded into zombies seeking their mother, "Are you my
mummy?" The Doctor aids in reprogramming the nanobots to cure the London population.
New Earth*Doctor*5000000023*5000000023*Tiger*New Earth*United States of America*New New New New New New
New New New New New New New New New York*Tyler*Rose***Sisters of Plenitude**The Sisters of Plenitude were
farming human clones for medical advancement. The Doctor frees the clones and establishes their
citizenship.
Tooth and Claw*Doctor*1879*1901-01-22*Wolf*Earth*Scotland*Aberdeenshire*of
Kent*Alexandrina*Victoria*Queen*Lupine Wavelength Haemovariform**Queen Victoria becomes a werewolf (is
infected by an alien spore/virus whose symptomology mimics a humanoid canine) and establishes an Empire of
the Wolf.
Blink*Doctor*2007*1920*Wolf*Earth*England*Hull*Sparrow*Sally***Weeping Angels*Innate ability of Weeping
Angels to displace beings in time*A small group of weeping angels prey on humans' time-potential energy.
With The Doctor's help, these weeping angels are tricked into looking at each other and are quantum locked
as statues forever.
The Vampires of Venice*Doctor*1580*1580*Demon*Earth*Italy*Venice*Pond*Amy***Saturnyns*Perception
filters*Refugees from The Silence, a family fled from Saturnyne to 16th century Venice to make a new home
for themselves -- while terraforming Venice. The Doctor saves the city but dooms the Saturnyns to
extinction.
The Hungry Earth*Doctor*2020*2020*God*Earth*Wales*Cwmtaff*Chaudhry*Nasreen***Silurians**A drilling company
discovers an underground civilization of ancient reptilian humanoids. The Doctor mediates talks and
prevents all out war between the Silurians and the humans.
Let's Kill Hitler*Doctor*1938*5100*Wolf*Earth*Germany*Berlin*Hitler*Adolf****Teselecta*The Doctor prevents
a rip in the time and space continuum by saving Hitler from being assassinated by a vigilante human crew
from the future.

```

Dark Water*Missy*2014-11-01*2014-11-08*God*Earth*England*London*Pink*Danny***Cybermen*Nethersphere
Matrix*Missy releases cybermen from the globally distributed 3W Institute mausoleums. Danny Pink takes
control of all Cyberman and commands all Cybermen to stand down and dismantle the Nethersphere.
;
run ;

APPENDIX C: L_HISTORY.SAS

```
** ----- ** ;
** Program:    l_history.sas          ** ;
** Author:     IMLeprince             ** ;
** Created:    2023-10-12             ** ;
**                                     ** ;
** Usage:      Generate a listings of Historical Events ** ;
**                                     ** ;
** REQUIRES srcdata:  EARTH           ** ;
** REQUIRES timedata: L_HISTORY_EARTH&PREVSTAMP ** ;
** REQUIRES macros:   TARDIS         ** ;
**                                     ** ;
** OUTPUT: L_HISTORY_EARTH.docx       ** ;
**         L_HISTORY_EARTH&TIMESTAMP.sas7bdat ** ;
**                                     ** ;
** Revisions:                                     ** ;
**                                     ** ;
** ----- ** ;

%*-----;
%* Setting up envirotnment: define the filepaths
%*-----;
/* SRCDATA   = location of source datasets (EARTH.sas7bdat)      */
/* TLFDATA   = location of derived datasets that support outputs */
/* TIMEDATA  = location of timestamped derived datasets that    */
/*            support data comparisons by TARDIS macro           */
/* OUTPUTS   = location of listing output (*.docx)               */

%let SRCDATA   = C:\Users\ileprince\Desktop\BASUG 2025\TLFs\Source Data ;
%let TLFDATA   = C:\Users\ileprince\Desktop\BASUG 2025\TLFs\Derived Data ;
%let TIMEDATA  = C:\Users\ileprince\Desktop\BASUG 2025\TLFs\Derived Data\TimeStamped ;
%let OUTPUTS   = C:\Users\ileprince\Desktop\BASUG 2025\TLFs\Outputs ;

libname srcdata "&srcdata" ;
libname TLFdata "&tlfdata" ;
libname timedata "&timedata" ;

%*-----;
%* Bring in data EARTH
%*-----;
```

```

data afile ;
  set srcdata.earth ;
  length asdtc aendtc $50 ;

  array datec stdatec endatec ;
  array years styear enyear ;
  array datei asdtc aendtc ;
  array daten asdt aendt ;

  do over datec ;
    years = input(scan(datec,1,'-'),best12.) ;
    datei = datec ;
    /* SAS dates are valid from January 1, 1582 to December 31, 20,000 */
    do while(1582 <= years <= 20000 and missing(scan(datei,3,'-'))) ;
      datei = cats(datei,'-01') ;
    end ;
    if not missing(scan(datei,3,'-')) then
      daten = input(datei,yymmdd10.) ;
    end ;

    if cmiss(stdatec,endatec) = 0 and stdatec ^= endatec then
      c_date = catx('/`n',stdatec,endatec) ;
    else c_date = strip(stdatec) ;

    c_human = catx(' ',human_title, human_first, human_middle, human_last) ;
    c_location = catx(', ',city, country) ;

    format asdt aendt yymmdd10. ;
run ;

proc sort data = afile ;
  by styear asdt enyear aendt timelord ;
run ;

%*-----;
%* Indicate New or Changed Data for Listing
%*-----;
%TARDIS (
  indata    = afile ,
  baseprog  = l_history ,

```

```

order      = _earth ,
sortvar    = styear astdt enyear aendt
            timelord c_human alien_race c_date c_location event alien_tech ,
arrayvar    = timelord c_human alien_race c_date c_location event alien_tech
) ;

%*-----;
%* Generate Listing
%*-----;

ods _all_ close ;
options orientation=landscape ;
ods escapechar='`' ;

title1      j=c "SAMPLE HISTORY LISTING USING TARDIS MACRO" ;
title2      j=l "Time Lord High Council"   j=r "BASUG 2025" ;
title3      j=c "Listing 1: History of Events" ;
title4      j=c "Planet Earth" ;

ods word file="%outputs\listing_history_earth.docx" ;
proc report data = afile missing split = '~' spacing=1 headskip headline spanrows ;
    column (timelord c_human alien_race c_date c_location event alien_tech) ;

    define timelord      / display "Time Lord"
                        center style=[width=0.60 in];
    define c_human       / display "Who~(Human)"
                        center style=[width=0.90 in] ;
    define alien_race    / display "What~(Alien Race)"
                        center style=[width=1.05 in] ;
    define c_date        / display "When~(Start Date/~End Date)"
                        center style=[width=1.25 in] ;
    define c_location    / display "Where~(Planet Earth)"
                        left   style=[width=1.05 in] ;
    define event         / display "Why~(Event Description)"
                        left   style=[width=3.50 in] ;
    define alien_tech    / display "How~(Relevant Alien Technology)"
                        left   style=[width=1.45 in] ;

run ;

ods _all_ close ;

```